COMPUTATIONS IN ALGEBRAIC GEOMETRY

GESSICA ALECCI AND MICHELE GRAFFEO

ABSTRACT. These notes follow the content of the course *"Computations in algebraic geometry"* held at *Unicamp IMECC* in Spring (Brazilian fall) 2025. The final goal of the course is to learn how to parameterise locally closed subset of the Hilbert schemes of points. During the course, four main topics have been covered in four lectures. These are classical birational geometry, deformations of monomial ideals, deformations of zero-dimensional ideals, Hilbert schemes of points. Each lecture contains explicit examples of computations via Macaulay2.

CONTENTS

Acknowledgements. We thank Pedro Santos for his help in writing the paper. The first author thanks Professor Marcos Jardim for the invitation to the IMECC/UNICAMP (Campinas - SP/BR).

1. Lecture 1

In this lecture we recall the definition of classical maps in algebraic geometry such as Veronese/Segre embeddings, projections and the standard Cremona transformation. In addition, there are examples in the text of how to perform calculations related to this topic using the software *Macaulay2*, [9].

1.1. **Notation.**

We work over the field of complex numbers $\mathbb{C}$.

- We adopt throughout the Zariski topology.
- We denote by R a polynomial ring with complex coefficients. If not specified, the set of variables will be
  - $x_0,\dots,x_n$ in the projective setting,
  - $x_1,\dots,x_n$ in the affine setting,
  - x,y in the affine two-dimensional setting.
- The ring R is endowed with the standard grading $\deg x_i = 1$, for all $i=1,\dots,n$. This gives $R=\oplus_{k\geq 0}R_k$ where

$$R_k = \left\{ f\in R \,\middle|\, f \text{ homogeneous, and } \deg f = k \right\}\cup\{0\}.$$

- We denote by $\mathbb{A}^n=\operatorname{Spec}(R)$ the n-dimensional affine space with coordinates (generators of R) $x_1,\dots,x_n$, if not specified otherwise.
- Similarly, $\mathbb{P}^n=\operatorname{Proj}(R)$ denotes the n-dimensional projective space with homogeneous coordinates (generators of R) $x_0,\dots,x_n$, if not specified otherwise.
- We denote by $H_i=\{x_i=0\}\subset\mathbb{P}^n$, for $i=0,\dots,n$, the i-th coordinate hyperplane. Moreover, we denote by $e_i\in\mathbb{P}^n$, for $i=0,\dots,n$, the i-th coordinate point.
- The symbols $\operatorname{Mat}(n,m,\mathbb{C})$ and $\operatorname{Sym}(n,\mathbb{C})\subset\operatorname{Mat}(n,n,\mathbb{C})$ denote respectively the vector spaces of matrices and symmetric matrices with complex entries. The symbol $\operatorname{GL}(n,\mathbb{C})\subset\operatorname{Mat}(n,n,\mathbb{C})$ denotes the general linear group.
- Given any subset $S\subset\mathbb{P}^n$, we denote by $\langle S\rangle\subset\mathbb{P}^n$ the smallest linear subspace containing S.
- $\operatorname{Bir}(X)$ denotes the group of birational transformations of a variety X.

1.2. **Veronese embeddings.**

The first non-trivial example of morphism between projective spaces is provided by the n-th Veronese embedding of degree d.

Definition 1.1. The *n-th Veronese embedding of degree d* is the morphism defined by

$$\begin{array}{ccc} \mathbb{P}^n & \xrightarrow{\quad \mathrm{v}_{n,d}\quad} & \mathbb{P}^{\binom{n+d}{d}-1} \\ [x_0:\cdots:x_n] & \longmapsto & \left[x_0^{\alpha_0}\cdots x_n^{\alpha_n}\mid \sum_{i=0}^n \alpha_i = n\right]. \end{array}$$

Proposition 1.2. *The morphism* $\mathrm{v}_{n,d}$ *is a closed immersion.*

Exercise 1.3. Prove Proposition 1.2. **Hint:** see Example 1.4.

Example 1.4. Let us explain in details the case $n=1$. Fix homogeneous coordinates x_0,x_1 on $\mathbb{P}^1$ and y_i, for $i=0,\dots,d$ on $\mathbb{P}^d$. The first-Veronese embedding of degree d reads then as

$$\begin{array}{ccc} \mathbb{P}^1 & \xrightarrow{\quad \mathrm{v}_{1,d}\quad} & \mathbb{P}^d \\ [x_0:x_1] & \longmapsto & \left[x_0^d : x_0^{d-1}x_1:\cdots:x_1^d\right]. \end{array}$$

Consider the charts $U_i = \{ x_i \neq 0 \} \cong \mathbb{A}^1$, for $i = 0,1$ and $V_j = \{ y_j \neq 0 \} \cong \mathbb{A}^d$, for $j = 0,\dots,d$. Then, on U_i we have coordinates

$$t_i = \left(\frac{x_1}{x_0}\right)^{(-1)^i},$$

for $i = 0,1$. The restrictions of $\mathrm{v}_{1,d}$ to U_0 and U_1 have the form

$$\begin{array}{ccc} U_0 & \longrightarrow & V_0 \\ t_0 & \longmapsto & (t_0, t_0^2, \dots, t_0^d), \end{array} \quad \text{and} \quad \begin{array}{ccc} U_1 & \longrightarrow & V_d \\ t_1 & \longmapsto & (t_1^d, t_1^{d-1}, \dots, t_1). \end{array} \tag{1.1}$$

In particular, the image of $\mathrm{v}_{1,d}$ is entirely contained $V_1 \cup V_d$. Notice also that the image is a smooth curve as both the maps in (1.1) are parametrisations of smooth curves.

Let us compute the image of the morphism $\mathrm{v}_{1,d}$. In order to do this, we first observe that any two consecutive entries of the map $\mathrm{v}_{1,d}$ have the same ratio, namely x_0/x_1. Therefore, the equality

$$[y_0 : \cdots : y_d] = \mathrm{v}_{1,d}([x_0 : x_1]) = \left[x_0^d : x_0^{d-1}x_1 : \cdots : x_1^d\right] \tag{1.2}$$

implies

$$\mathrm{rk}\begin{bmatrix} y_0 & \cdots & y_{d-1} \\ y_1 & \cdots & y_d \end{bmatrix} \leq 1.$$

Define $Y_{1,d} \subset \mathbb{P}^d$ to be the closed subset given by

$$Y_{1,d} = \left\{ [y_0 : \cdots : y_d] \in \mathbb{P}^d \;\middle|\; \mathrm{rk}\begin{bmatrix} y_0 & \cdots & y_{d-1} \\ y_1 & \cdots & y_d \end{bmatrix} \leq 1 \right\}.$$

Clearly $\mathrm{v}_{1,n}(\mathbb{P}^1) \subset Y_{1,d}$. We prove that $\mathrm{v}_{1,d} : \mathbb{P}^1 \to Y_{1,d}$ is invertible and this implies that $\mathrm{v}_{1,d}$ is a closed immersion. Its inverse is

$$\begin{array}{ccc} Y_{1,d} & \xrightarrow{\rho_{1,d}} & \mathbb{P}^1 \\ [y_0 : \cdots : y_d] & \longmapsto & [y_0 : y_1]. \end{array}$$

To see this we only have to show that the map $\rho_{1,d}$ is well defined. This is true because it can be extended to points of the form $[0 : 0 : y_2 : \cdots : y_d] \in Y_{1,d}$ via

$$[y_0 : \cdots : y_d] \mapsto [y_{d-1} : y_d].$$

Indeed, the conditions $y_0 = y_1 = 0$ and $y_{d-1} = y_d = 0$ are incompatible on $Y_{1,d}$ and $[y_0 : y_1] = [y_{d-1} : y_d]$, by (1.2).

Definition 1.5. The rational normal curve of degree d is the image of the morphism $\mathrm{v}_{1,d}$. If $d = 2$, the rational normal curve is called conic, and for $d = 3$, it is called twisted cubic.

Example 1.6. We describe now the degree-2 Veronese embeddings. In this setting the projective space $\mathbb{P}^{\binom{n+2}{2}-1}$ identifies with $\mathbb{P}\,\mathrm{Sym}(n+1,\mathbb{C})$ and we get the following expression for the morphism $\mathrm{v}_{n,2}$.

$$\begin{array}{ccc} \mathbb{P}^n & \xrightarrow{\mathrm{v}_{n,2}} & \mathbb{P}\,\mathrm{Sym}(n+1,\mathbb{C}) \\ [x_0 : \cdots : x_n] & \longmapsto & \begin{bmatrix} x_0^2 & \cdots & \cdots & x_0 x_n \\ x_1 x_0 & x_1^2 & \cdots & \vdots \\ \vdots & \vdots & \ddots & \vdots \\ x_0 x_n & x_1 x_n & \cdots & x_n^2 \end{bmatrix}. \end{array}$$

Exercise 1.7. Prove that

$$\mathrm{v}_{n,2}(\mathbb{P}^n) = \left\{ [M] \in \mathbb{P}\mathrm{Sym}(n+1,\mathbb{C}) \,\middle|\, \mathrm{rk}\, M \le 1 \right\}. \tag{1.3}$$

Hint: Clearly $\mathrm{v}_{n,2}(\mathbb{P}^n)$ is contained in the right hand side of Equation (1.3). For the opposite inclusion consider the projections on the i-th row (or the i-th column), for $i = 0,\dots,n$.

Example 1.8. The easiest example is the second Veronese surface, i.e. the image of the morphism

$$\begin{array}{ccc} \mathbb{P}^2 & \xrightarrow{\mathrm{v}_{2,2}} & \mathbb{P}\mathrm{Sym}(3,\mathbb{C}) \cong \mathbb{P}^5 \\ [x_0 : x_1 : x_2] & \longmapsto & \begin{bmatrix} x_0^2 & x_0x_1 & x_0x_2 \\ x_1x_0 & x_1^2 & x_1x_2 \\ x_2x_0 & x_2x_1 & x_2^2 \end{bmatrix}. \end{array}$$

One of the many possible instances in which Veronese embeddings turn out to be useful is the description of loci parametrising homogeneous ideals.

Exercise 1.9. Fix some integer $d > 1$. Describe the locus

$$V_d^{\mathrm{o}} = \left\{ [f] \in \mathbb{P}R_d \,\middle|\, \dim_{\mathbb{C}} \mathrm{Span}\left(\frac{\partial}{\partial x_i} f \right)_{i=0}^{n} > 1 \right\}.$$

We get

$$V_d^{\mathrm{o}} = \mathbb{P}R_d \setminus \mathrm{v}_{n,d}(\mathbb{P}^n).$$

Exercise 1.10. Two important aspects of the Veronese embedding concern its degree and its projective normality.

- Let $L \subset \mathbb{P}^n$ be a line. Then, we have $\mathrm{v}_{n,d}|_L \equiv \mathrm{v}_{1,d}$. Moreover, the generic hyperplane intersects $\mathrm{v}_{n,d}(L)$ in d distinct points. We say that the morphism has topological degree d.
- Let $Y_d \subset \mathbb{P}^n$ be a hypersurface of degree d, i.e. $Y_d = V(f)$ for some $f \in R_d$. Then there exists a hyperplane $H \subset \mathbb{P}^{\binom{n+d}{d}-1}$ such that $Y_d = \mathrm{v}_{n,d}(\mathbb{P}^n) \cap H$. We say that $\mathrm{v}_{n,d}$ is projectively normal.

Remark 1.11. It is worth mentioning that topological degree and projective normality are defined in a much more general context. See [8] for more details on the topological degree and [10, §I.3, §II.5] for projective normality.

The following exercise shows that not all degree d closed immersions are projectively normal.

Exercise 1.12. Show that the composition

$$\begin{array}{ccccc} \mathbb{P}^1 & \longrightarrow & \mathbb{P}^4 & \dashrightarrow & \mathbb{P}^3 \\ [x_0 : x_1] & \longmapsto & [x_0^4 : x_0^3x_1 : x_0^2x_1^2 : x_0x_1^3 : x_1^4] & \longmapsto & [x_0^4 : x_0^3x_1 : x_0x_1^3 : x_1^4] \end{array}$$

is a closed immersion of topological degree 4. Show that it is not projective normal.

1.2.1. *Example of computation.*

```
-- Define a function that remove the common factors from the entries of a map
cleanFactors = f -> (
    L=(entries matrix f)#0;
    G=gcd L;
    Laux= for l in L list sub(l/G,source f );
    return map(target f, source f, Laux);
)
```

```
-- Declear ambient spaces
n = 3; d = 3; N = binomial(n+d,d)-1; R = QQ[x_0 .. x_n]; S = QQ[y_0 .. y_N];
-- Construct the embedding and compute the equations
vnd1 = ( ideal R_* )^d_*;                              -- 1st possibility
vnd2 = (entries(monomials (sum R_*)^d))#0;             -- 2nd possibility
set vnd1 == set vnd2                                   -- Check the two are the same
Verond = map(R,S,vnd1);                                -- Define the map
Equations = trim ker Verond;                           -- Compute equations
-- Check of the smoothness of the image of e_0={1,0,0,..,0}.
-- WLOG we restrict to { x_0 != 0 } and { y_0 != 0 }.
-- For simplicity we keep the same symbols for the variables
R0 = QQ[ x_1 .. x_n ]; S0 = QQ[ y_1 .. y_N ];
vnd10 = drop(for v in vnd1 list sub(sub( v , x_0=>1),R0),1);
Verond0 = map(R0,S0,vnd10  );
Eqs0 = trim ker Verond0;
Eqs0 == sub(sub( Equations,y_0=>1),S0)               -- It agrees with dehomogenising
J = sub( jacobian Eqs0, for i from 1 to N list y_i =>0 ); -- Jacobian at e_0
0 == (rank J - N + n)                                -- Check it has rank N - n
-- Computation of the inverse map
use R; use S;
Inverse = map(S,R,for i from 0 to n list y_i);
cleanFactors ( Verond * Inverse )           -- Check that the composition is the identity
```

1.3. **Segre embeddings.** Segre embeddings provide a way to realise products of projective spaces as closed subsets of an ambient projective space.

Definition 1.13. Given two integers $n, m \in \mathbb{Z}_{\geq 0}$, the Segre (n,m)-embedding is the morphism

$$\begin{array}{ccc} \mathbb{P}^n \times \mathbb{P}^m & \xrightarrow{\mathrm{s}_{n,m}} & \mathbb{P}\mathrm{Mat}(n+1, m+1, \mathbb{C}) \cong \mathbb{P}^{(n+1)(m+1)-1} \\ ([x_0 : \cdots : x_n], [y_0 : \cdots : y_m]) & \longmapsto & \begin{bmatrix} x_0 y_0 & \cdots & x_0 y_m \\ \vdots & \ddots & \vdots \\ x_n y_0 & \cdots & x_n y_m \end{bmatrix} \end{array}$$

Proposition 1.14. *The morphism* $\mathrm{s}_{n,m}$ *is a closed immersion.*

Exercise 1.15. Prove Proposition 1.14.
Hint: Define

$$S_{n,m} = \left\{ [M] \in \mathbb{P}^{(n+1)(m+1)-1} \,\middle|\, \mathrm{rk}\, M \leq 1 \right\}.$$

Clearly, we have

$$\mathrm{s}_{n,m}(\mathbb{P}^n \times \mathbb{P}^m) \subset S_{n,m}.$$

To conclude consider the morphism

$$\begin{array}{ccc} S_{n,m} & \longrightarrow & \mathbb{P}^n \times \mathbb{P}^m \\ \begin{bmatrix} z_{0,0} & \cdots & z_{0,m} \\ \vdots & \ddots & \vdots \\ z_{n,0} & \cdots & z_{n,m} \end{bmatrix} & \longmapsto & ([z_{0,0} : \cdots : z_{n,0}], [z_{0,0} : \cdots : z_{0,m}]). \end{array}$$

Example 1.16. For $n = m = 1$, we get

$$\begin{array}{ccc} \mathbb{P}^1 \times \mathbb{P}^1 & \xrightarrow{\quad \mathrm{s}_{1,1} \quad} & \mathbb{P}^3 \\ ([x_0 : x_1],[y_0 : y_1]) & \longmapsto & \begin{bmatrix} x_0 y_0 & x_0 y_1 \\ x_1 y_0 & x_1 y_1 \end{bmatrix} \end{array}$$

and

$$\mathrm{s}_{1,1}(\mathbb{P}^1 \times \mathbb{P}^1) = \left\{ z_{0,0}z_{1,1} - z_{1,0}z_{0,1} = 0 \right\}.$$

Since all smooth quadrics of $\mathbb{P}^n$ differ by projectivities, we deduce that all smooth quadrics of $\mathbb{P}^3$ are isomorphic to $(\mathbb{P}^1)^{\times 2}$.

Example 1.17. Consider the diagonal

$$\Delta = \left\{ (p,q) \in \mathbb{P}^n \times \mathbb{P}^n \;\middle|\; p = q \right\} \cong \mathbb{P}^n.$$

We then restrict the Segre embedding to $\Delta \subset \mathbb{P}^n \times \mathbb{P}^n$ and we get the second Veronese embedding $\mathrm{v}_{n,2} : \Delta \to \mathbb{P}\,\mathrm{Sym}(n+1,\mathbb{C})$. Precisely, the following diagram

$$\begin{tikzcd} \Delta \arrow[rr, "\mathrm{s}_{n,n}|_\Delta"] & & \mathbb{P}^{(n+1)^2-1} \\ {[x_0 : \cdots : x_n]} \arrow[rr, mapsto] \arrow[dr, "\mathrm{v}_{n,2}"'] & & \begin{bmatrix} x_0^2 & \cdots & \cdots & x_0x_n \\ x_1x_0 & x_1^2 & \cdots & \vdots \\ \vdots & \vdots & \ddots & \vdots \\ x_0x_n & x_1x_n & \cdots & x_n^2 \end{bmatrix} \\ & \mathbb{P}\,\mathrm{Sym}(n+1,\mathbb{C}), \arrow[ur, hook] & \end{tikzcd}$$

commutes.

1.4. **Projections and blowups of linear subspaces.**

Definition 1.18. Let X and Y be quasi-projective varieties. A rational map $\varphi : X \dashrightarrow Y$ is an equivalence class of pairs (U, f), where U is a dense open subset of X and f is a morphism from U to Y, and where two pairs (U_1, f_1) and (U_2, f_2) are equivalent if there exists a dense open subset $V \subset U_1 \cap U_2$ such that $f_1|_V = f_2|_V$.

Projections are the first examples of rational maps.

Let $H, K \subset \mathbb{P}^n$ be linear subspaces such that $H \cong \mathbb{P}^k$, $K \cong \mathbb{P}^{n-k-1}$ and $H \cap K = \varnothing$. Without loss of generality we put

$$K = \{ x_0 = \cdots = x_k = 0 \},\ H = \{ x_{k+1} = \cdots = x_n = 0 \} \subset \mathbb{P}^n.$$

The projection onto H with centre K is the rational map

$$\begin{array}{ccc} \mathbb{P}^n & \overset{\pi_K}{\dashrightarrow} & H \\ {[x_0 : \cdots : x_n]} & \longmapsto & [x_0 : \cdots : x_k]. \end{array} \tag{1.4}$$

Notice that we do not report the dependence on H as it is not useful for our applications and we prefer to keep the notation as simple as possible.

Remark 1.19. Geometrically, for all points $p \in \mathbb{P}^n \setminus K$ we consider $W_p = \langle p, K \rangle$. Then we have

$$\pi_K(p) = W_p \cap H.$$

Exercise 1.20. Prove the following basic properties of π_K.

- Any linear subspace $W \subset \mathbb{P}^n$ such that $K \subset W$ and $W \cong \mathbb{P}^{n-k}$ is contracted to a point, i.e. $\pi_K(W) = p \in H$.
- If $L \subset \mathbb{P}^n$ is a line then $\pi_K(L) = p$ is a point if and only if $L \cap K \neq \emptyset$.[1]

Definition 1.21. Given a rational map

$$\begin{array}{ccc} \mathbb{P}^n & \overset{\varphi}{\dashrightarrow} & \mathbb{P}^m \\ [x_0 : \cdots : x_n] & \longmapsto & [f_i([x_0 : \cdots : x_n]) \mid i = 0, \dots, m], \end{array}$$

where $f_i \in R$, for $i = 0, \dots, m$, denote by $\operatorname{dom}(\varphi)$ its maximal domain of definition and by $\operatorname{Ind}(\varphi)$ its indeterminacy locus. In symbols[2]

$$\operatorname{dom}(\varphi) = \bigcup_{(U,f)\in\varphi} U, \text{ and } \operatorname{Ind}\varphi = \operatorname{Proj}(R/(f_0, \dots, f_m)).$$

Then, the graph of φ is

$$\operatorname{graph}\varphi = \left\{ (p,q) \in \operatorname{dom}(\varphi) \times \mathbb{P}^m \;\middle|\; q = \varphi(p) \right\} \subset \mathbb{P}^n \times \mathbb{P}^m.$$

Exercise 1.22. Let $\varphi : \mathbb{P}^n \dashrightarrow \mathbb{P}^m$ be a rational map, and denote by $Z = \operatorname{Ind}\varphi$ its indeterminacy locus. Consider the following diagram

$$\begin{tikzcd} & \overline{\operatorname{graph}\varphi} \arrow[dl, "\varepsilon_Z"'] \arrow[dr, "\pi"] & \\ \mathbb{P}^n \arrow[rr, dashed, "\varphi"] & & \mathbb{P}^m, \end{tikzcd}$$

where the closure $\overline{\operatorname{graph}\varphi}$ is taken in $\mathbb{P}^n \times \mathbb{P}^m$, the morphisms ε_Z and π are the restrictions of the canonical projections to $\overline{\operatorname{graph}\varphi} \subset \mathbb{P}^n \times \mathbb{P}^m$.

Prove the following properties.

- The first projection ε_Z is birational. (**Hint:** Look at its restriction to $\operatorname{graph}\varphi$.)
- Suppose that φ is the projection π_K in (1.4). In particular $Z = K$ is a linear subspace. Show that
  - if $\dim K = n-1$, then ε_K is an isomorphism,
  - the morphism π is a fibration with fibres isomorphic to $\mathbb{P}^{n-k}$,
  - the preimage $E_K = \varepsilon_K^{-1} K$ is a $\mathbb{P}^{n-k-1}$-bundle over K.

Definition 1.23. We say that ε_Z is the blowup of $\mathbb{P}^n$ with centre Z. We will also call blowup the variety $\overline{\operatorname{graph}\varphi}$ ad we will denote it by $\operatorname{Bl}_Z \mathbb{P}^n$. Finally we say that E_Z is the exceptional locus.

Exercise 1.24. Definition 1.23 is well posed, i.e. it does not depend on φ but only on $\operatorname{Ind}\varphi$.

Precisely, show that given two rational maps $\varphi_i : \mathbb{P}^n \dashrightarrow \mathbb{P}^{n_i}$, for $i = 1, 2$ having the same indeterminacy locus $\operatorname{Ind}\varphi_1 = \operatorname{Ind}\varphi_2$ the two closures $\overline{\operatorname{graph}\varphi_i}$, for $i = 1, 2$, are canonically isomorphic.

Exercise 1.25. Extend Definition 1.21, Exercise 1.22 and Definition 1.23 to the following cases.

- Products of projective spaces.
- Arbitrary quasi-projective varieties.

[1] In particular, we require $L \not\subset K$ so that it makes sense to consider $\pi_K(L)$.

[2] Recall that by definition a rational map is an equivalence class. Thus it is a set.

Example 1.26 (Blowup at a point)**.** We describe now the projection π_K in the case $n=2$ and $\dim K=0$. Without loss of generality we put $K=\{e_2\}$ and we consider the projection

$$\begin{array}{ccc} \mathbb{P}^2 & \overset{\pi_{e_2}}{\dashrightarrow} & \mathbb{P}^1 \\ [x_0:x_1:x_2] & \longmapsto & [x_0:x_1], \end{array}$$

from $\mathbb{P}^2$ with centre the coordinate point $e_2=[0:0:1]$. Then, the blowup with centre e_2 is

$$\begin{aligned} \mathrm{Bl}_{e_2}\mathbb{P}^2 &= \overline{\mathrm{graph}(\pi_{e_2})} \\ &= \overline{\left\{ (p,q)\in \mathrm{dom}(\pi_{e_2})\times\mathbb{P}^1 \;\middle|\; \pi_{e_2}(p)=q \right\}} \subset \mathbb{P}^2\times\mathbb{P}^1 \\ &= \overline{\left\{ ([x_0:x_1:x_2],[y_0:y_1])\in \mathrm{dom}(\pi_{e_2})\times\mathbb{P}^1 \;\middle|\; \det\begin{pmatrix} x_0 & x_1 \\ y_0 & y_1 \end{pmatrix}=0 \right\}} \subset \mathbb{P}^2\times\mathbb{P}^1 \\ &= \left\{ ([x_0:x_1:x_2],[y_0:y_1])\in \mathbb{P}^2\times\mathbb{P}^1 \;\middle|\; x_0y_1-x_1y_0=0 \right\} \subset \mathbb{P}^2\times\mathbb{P}^1. \end{aligned}$$

We stress that there is a commutative diagram

$$\begin{tikzcd} & \mathrm{Bl}_{e_2}\mathbb{P}^2 \subset \mathbb{P}^2\times\mathbb{P}^1 \arrow[dl, "\pi_2|_{\mathrm{Bl}\,\mathbb{P}^2}"'] \arrow[dr, "\pi_1|_{\mathrm{Bl}_{e_2}\mathbb{P}^2}"] & \\ \mathbb{P}^2 \arrow[rr, dashed, "\pi_{e_2}"'] & & \mathbb{P}^1, \end{tikzcd}$$

where $\pi_2|_{\mathrm{Bl}\,\mathbb{P}^2}$ is the blowup morphism and $\pi_1|_{\mathrm{Bl}_{e_2}\mathbb{P}^2}$ is a $\mathbb{P}^1$-fibration as described in Exercise 1.22.

Exercise 1.27. Compute the equations of some embeddings of $\mathrm{Bl}_{\mathbb{P}^k}\mathbb{P}^n$.

Example 1.28 (Blowup at two points)**.** We construct now the blowup of the projective plane at two points as the closure of a birational map given by a pair of projections. Consider the rational map

$$\begin{array}{ccc} \mathbb{P}^2 & \overset{\varphi}{\dashrightarrow} & \mathbb{P}^1\times\mathbb{P}^1 \\ [x_0:x_1:x_2] & \longmapsto & ([x_0:x_1],[x_1:x_2]). \end{array}$$

It is a birational map with inverse

$$\begin{array}{ccc} \mathbb{P}^1\times\mathbb{P}^1 & \overset{\psi}{\dashrightarrow} & \mathbb{P}^2 \\ ([y_0:y_1],[z_0:z_1]) & \longmapsto & [y_0z_0:y_1z_0:y_1z_1]. \end{array}$$

The two maps have the following indeterminacy loci

$$\mathrm{Ind}(\varphi)=\{e_0,e_2\} \quad \text{and} \quad \mathrm{Ind}(\psi)=\{([1:0],[0:1])\}.$$

Figure 1 depicts the construction.



FIGURE 1. Pictorial description of the construction.

As a consequence, we get the isomorphism

$$\mathrm{Bl}_{p,q}\mathbb{P}^2 \cong \mathrm{Bl}_r(\mathbb{P}^1 \times \mathbb{P}^1),$$

where $p, q \in \mathbb{P}^2$ and $r \in \mathbb{P}^1 \times \mathbb{P}^1$.

Exercise 1.29. Study the blowup of the projective plane at two points.

- Check the details in Example 1.28.
- Compute the equations of $\mathrm{Bl}_{e_0,e_2}\mathbb{P}^2 \subset \mathbb{P}^2 \times \mathbb{P}^1 \times \mathbb{P}^1$.
- Show that φ contracts the line trough e_0 and e_2 on the unique point of $\mathrm{Ind}\,\psi$.
- Show that ψ contracts the two lines trough the unique point of $\mathrm{Ind}\,\psi$ on e_0 and e_2.

The following exercise is a direct generalisation of the construction in Example 1.28.

Exercise 1.30. Let $n = 2k+1$ be an odd nonnegative integer and let $Q_n \subset \mathbb{P}^n$ be a smooth quadric. Let also $Q_{n-2} \subset \mathbb{P}^{n-2} \subset \mathbb{P}^{n-1}$ be a smooth quadric. Prove that there is an isomorphism

$$\mathrm{Bl}_p Q_n \cong \mathrm{Bl}_{Q_{n-2}}\mathbb{P}^{n-1}.$$

1.4.1. *Example of computation: use the command* `normalCone` *for the blowup.*

```
    cleanList = L -> (
    G=gcd L;
    if G!= 0 then return for l in L list sub(l/G,ring L#0 );
    return L
)
-- Declear ambient space
R = QQ[x_0 .. x_2];
S = R[w_0 , w_1];
projection = {x_1,x_0};
I = minors (2,matrix {projection ,S_* });
N = normalCone ideal projection ;
K =sub(ker map(ambient N , ambient ambient N ),S)
I==K                                              -- Check the construction is the same
O = QQ [R_*|S_*]
point={for r in R_* list sub(r,O),for s in S_* list sub(s,O)};
-- Function to pass from ideal to parametrisation
evalIdeal = I -> for u in point list cleanList apply( u , v -> sub(sub(v,O/I),O) ) ;
E = sub(ideal projection , O ) + sub(I,O)                 -- Preimage of the origin
```

```
evalIdeal E
F = saturate(sub(ideal(3*w_0 -w_1),0)+sub(I,0),sub(ideal S_*,0))                 -- Fibration
evalIdeal F
```

1.5. Standard Cremona transformation.

The standard Cremona transformation $\mathrm{c}_n \in \mathrm{Bir}(\mathbb{P}^n)$, for $n \geq 1$, is a birational transformation of the projective space $\mathbb{P}^n$ and it is the first non trivial example of birational map. We recall now its definition.

Definition 1.31. The standard Cremona transformation $\mathrm{c}_n \in \mathrm{Bir}(\mathbb{P}^n)$ is

$$
\begin{array}{ccc}
\mathbb{P}^n & \overset{\mathrm{c}_n}{\dashrightarrow} & \mathbb{P}^n \\
[x_0 : \cdots : x_n] & \longmapsto & \left[\frac{1}{x_0} : \cdots : \frac{1}{x_n}\right] = [x_1 \cdots x_n : \cdots : x_0 \cdots x_{n-1}].
\end{array}
$$

In dimension $n = 2$ the standard Cremona transformation plays a special role as the following classical result explains.

Theorem 1.32 (Noether-Castelnuovo, [5, 12])**.** *The following equality of groups holds true,*

$$\mathrm{Bir}(\mathbb{P}^2) = \langle \mathbb{P}\,\mathrm{GL}(3,\mathbb{C}), \mathrm{c}_2 \rangle.$$

Exercise 1.33. Prove the main properties of the standard Cremona transformation listed below.

- $\mathrm{c}_n^2 \equiv \mathrm{id}_{\mathbb{P}^n}$
- $\mathrm{Ind}\,\mathrm{c}_n = \coprod_{0 \leq i < j \leq n} \{ x_i = x_j = 0 \}$
- $\mathrm{c}_n(\{ x_i = 0 \}) = e_i$
- $\mathrm{Fix}\,\mathrm{c}_n = \{ [1 : \pm 1 : \pm 1 : \cdots : \pm 1] \}$
- $|\mathrm{Fix}\,\mathrm{c}_n| = 2^n$

1.5.1. *Example of computation.*

```
-- Define a function that remove common factor from lists
cleanList = L -> (
    G=gcd L;
    if G!= 0 then return for l in L list sub(l/G,ring L#0 );
    return L
)
-- Function to pass from ideal to parametrisation
evalIdeal = I -> cleanList for l in R_* list sub(sub(l,R/I),R);
-- Declear ambient space
n = 3;   R = QQ[x_0 .. x_n];
Crent = for i from 0 to n list product for j in delete (i, 0..n ) list x_j;
Cremona = map(R , R , Crent );                             -- Define the map
cleanFactorsMap ( Cremona * Cremona )                      -- Check it is an involution
DDD= set primaryDecomposition ideal Crent                  -- Compute the indeterminacy locus
-- Check that the i-th coord. hyperplane is contracted to the i-th coord. point
for i from  0 to n do print cleanList for a in Crent list  sub( a  , x_i=>0)
DDD1 = set primaryDecomposition  minors(2,matrix{R_*,Crent});
DDD1=DDD1-DDD                                              -- Fixed points
for d in toList DDD1 list evalIdeal d
```

Exercise 1.34 (Blowup at three points)**.** Consider the standard Cremona transformation $\mathrm{c}_2 \in \mathrm{Bir}(\mathbb{P}^2)$. We have a commutative diagram

$$\begin{tikzcd} & B \arrow[dl, "\varepsilon"'] \arrow[dr, "\theta"] & \\ \mathbb{P}^2 \arrow[rr, dashed, "\mathrm{c}_2"] & & \mathbb{P}^2 \\ {[x_1 : x_2 : x_3]} \arrow[rr, mapsto] & & \left[\frac{1}{x_1} : \frac{1}{x_2} : \frac{1}{x_3}\right], \end{tikzcd} \tag{1.5}$$

where $B = \overline{\mathrm{graph}(\mathrm{c}_2)}$ is the closure of the graph.

- Realise the blowup $\mathrm{Bl}_{e_0,e_1,e_2}\mathbb{P}^2$ as a closed subset of $\mathbb{P}^2 \times \mathbb{P}^2$, see Figure 2.
- Realise the blowup $\mathrm{Bl}_{e_0,e_1,e_2}\mathbb{P}^2$ as a closed subset of $\mathbb{P}^2 \times \mathbb{P}^1 \times \mathbb{P}^1 \times \mathbb{P}^1$.
- Show that the restriction of the canonical projection

  $$\mathbb{P}^2 \times \mathbb{P}^1 \times \mathbb{P}^1 \times \mathbb{P}^1 \to \mathbb{P}^1 \times \mathbb{P}^1 \times \mathbb{P}^1$$

  to $\mathrm{Bl}_{e_0,e_1,e_2}\mathbb{P}^2$ is a closed immersion.



FIGURE 2. The resolution of the indeterminacies of the standard Cremona transformation in dimension 2.

Exercise 1.35. Let $T = \bigcup\limits_{0 \le i < j \le 3} \left\{ x_i = x_j = 0 \right\} \subset \mathbb{P}^3$ be the coordinate tetrahedron, i.e. the union of the coordinate lines of $\mathbb{P}^3$. Denote by X the blowup $X = \mathrm{Bl}_T \mathbb{P}^3$.

- Realise X as a closed subset of $\mathbb{P}^3 \times \mathbb{P}^3$.
- Show that X has 12 singular points.
- Identify all the irreducible components of the exceptional locus E_T. (**Hint:** there are 10 of them.)

```
-- Declear ambient space
R = QQ[x_0 .. x_3];
S = R[w_0 .. w_3];

Crent = for i from 0 to 3 list product for j in delete (i, 0..3 ) list x_j;
D = minimalPrimes minors (2,matrix {Crent ,S_* });
I = (select(D , d-> dim d<6 ))#0;
E= normalCone ideal Crent;
K=sub(ker map(ambient E , ambient ambient E),S)
I==K                                                -- Check the construction is the same
O = QQ [R_*|S_*]
point={for r in R_* list sub(r,O),for s in S_* list sub(s,O)};
-- Function to pass from ideal to parametrisation
```

```
evalIdeal = I ->  for u in point list cleanList  apply( u , v ->  sub(sub(v,0/I),0) ) ;
I = sub(I,0);
J = jacobian I;
U = saturate(saturate(trim(minors (3,J)+I),ideal(x_0..x_3)), ideal(w_0..w_3));
for u in minimalPrimes U list evalIdeal u
```

2. Lecture 2

In this second lecture, we introduce some useful tools for dealing with ideals of the polynomial ring, such as the notion of socle and Macaulay duality. Along the way, we will see many examples concerning monomial ideals. As a computational example, we treat the combinatorial problem of counting higher-dimensional partitions. Finally, we present the first examples of deformation theory in terms of the module $\mathrm{Hom}_R(I,R/I)$.

2.1. Notation.

We work over the field of complex numbers $\mathbb{C}$.

- By algebra (resp. A-module) we implicitly mean $\mathbb{C}$-algebra (resp. A-module) of finite type over $\mathbb{C}$ (resp. over A).
- We denote by R a polynomial ring with complex coefficients and by $\mathfrak{m}\subset R$ the maximal ideal generated by the variables. If not specified, the set of variables will be
  - $x_1,\dots,x_n$ in the n-dimensional setting,
  - x,y in the two-dimensional setting.
- The ring R is endowed with the standard grading $\deg x_i = 1$, for all $i=1,\dots,n$. This gives $R=\oplus_{k\geq 0}R_k$ where
$$R_k=\left\{ f\in R \,\middle|\, f \text{ homogeneous, and } \deg f = k\right\}\cup\{0\}.$$
- We will denote by the same symbols the variables in R and their image in the quotient R/I.
- Given a $\mathfrak{m}$-primary ideal I, we denote by $\operatorname{len} R/I$ its colength, i.e.
$$\operatorname{len} R/I=\dim_{\mathbb{C}} R/I.$$
- The semigroup $\mathbb{N}^n$ is endowed with the poset structure given by componentwise comparison. All its subsets will be considered as poset with the restricted structure.

2.2. Graded modules and Hilbert–Samuel function.

Definition 2.1. An algebra A is graded if there exists a direct sum decomposition $A=\bigoplus_{k\in\mathbb{Z}}A_k$ such that the multiplication on A reads as follows in each degree $k\in\mathbb{Z}$,
$$\begin{aligned} A_h\times A_k &\longrightarrow A_{h+k}\\ (a,b) &\longmapsto ab.\end{aligned}$$
Given a graded algebra $A=\bigoplus_{k\in\mathbb{Z}}A_k$, an A-module M is graded if $M=\bigoplus_{k\in\mathbb{Z}}M_k$ and the action of A on M restricts as follows
$$A_h\times M_k \longrightarrow M_{h+k},$$
for all $h,k\in\mathbb{Z}$.

Definition 2.2. Let $(A,\mathfrak{m}_A)$ be a local algebra. Its associated graded algebra $\mathscr{G}\mathrm{r}_{\mathfrak{m}_A}A$ is the algebra
$$\mathscr{G}\mathrm{r}_{\mathfrak{m}_A}A=\bigoplus_{i\geq 0}\mathfrak{m}_A^i/\mathfrak{m}_A^{i+1}.$$

Exercise 2.3. Prove the following properties.

- The associated graded algebra $\mathscr{G}\mathrm{r}_{\mathfrak{m}_A} A$ is graded.
- Recall that a homogeneous ideal $I \subset R$ is an ideal which admits a generating set containing only homogeneous elements. Let I be a homogeneous ideal. Show that
  - I is a graded R-module,
  - R/I is a graded R-module,
  - $\operatorname{Hom}_R(I, R/I)$ is a graded module with
  $$\operatorname{Hom}_R(I,R/I)_d = \left\{ \varphi \in \operatorname{Hom}_R(I,R/I) \,\middle|\, \varphi(I_i) \subset (R/I)_{i+d}, \text{ for all } i \in \mathbb{Z} \right\} \cup \{0\}.$$
  Notice that, even if $I_d \cong (R/I)_d \cong (0)$ for all $d < 0$, the same is not necessarily true for the module $\operatorname{Hom}_R(I, R/I)$.

Definition 2.4. Consider an element $f \in R$ and write it as
$$f = \sum_{i=k_f}^{\deg(f)} f_i,$$
where $f_i \in R_i$, for $i = k_f, \dots, \deg(f)$. Then, the initial form of f is $\operatorname{In} f = f_{k_f}$. Moreover, if $I \subset R$ is any ideal, its initial ideal is
$$\operatorname{In} I = \left\{ \operatorname{In} f \,\middle|\, f \in I \right\}.$$

Exercise 2.5. Let $I \subset R$ be any ideal. Then,

- the initial ideal $\operatorname{In} I$ is homogeneous,
- if I is $\mathfrak{m}$-primary, $\mathscr{G}\mathrm{r}_{\mathfrak{m}/I} R/I \cong R/\operatorname{In} I$.

Example 2.6. Let $f = y^2 - x^2(x+1)$ be the polynomial defining the nodal cubic. Then, we have $\operatorname{In} f = y^2 - x^2$. While, for the ideal $I = (x + y^2 + y^3 + y^4, y^{10})$, we have $\operatorname{In}(I) = (x, y^{10})$.

Exercise 2.7. If the scheme $X = \operatorname{Spec}(R/I)$ is smooth at the origin $0 \in X \subset \mathbb{A}^n$, then we have an isomorphism of $\mathbb{A}^n$-schemes
$$\operatorname{Spec} R/\operatorname{In} I \cong \operatorname{Spec} \operatorname{Sym}(\overline{\mathfrak{m}}/\overline{\mathfrak{m}}^2),$$
where $\overline{\mathfrak{m}} = \mathfrak{m}/I$.

Remark 2.8. In general, the initial ideal defines the so-called tangent cone to a variety $X \subset \mathbb{A}^n$ at the origin $0 \in \mathbb{A}^n$. Roughly speaking it is the union of the lines having the maximum possible multiplicity intersection with X at the origin, see [6, §5.4] for more details.

Definition 2.9. Let A be a graded $\mathbb{C}$-algebra of finite type, and let $M = \bigoplus_{k \in \mathbb{Z}} M_k$ be a finitely generated graded A-module. The Hilbert–Samuel function associated to M is
$$\begin{aligned} \mathbb{Z} &\xrightarrow{h_M} \mathbb{N} \\ i &\longmapsto \dim_{\mathbb{C}} M_i. \end{aligned}$$
Now, let $(A, \mathfrak{m}_A)$ be an Artinian local $\mathbb{C}$-algebra of finite type. The Hilbert–Samuel function associated to A is
$$\begin{aligned} \mathbb{Z} &\xrightarrow{h_A} \mathbb{N} \\ i &\longmapsto \dim_{\mathbb{C}} {\mathfrak{m}_A}^i/{\mathfrak{m}_A}^{i+1}. \end{aligned}$$

Exercise 2.10. Definition 2.9 is well posed, i.e. the vector spaces M_k have finite dimension over $\mathbb{C}$, for all $k \in \mathbb{Z}$.

Remark 2.11. Notice that

$$h_A \equiv h_{\mathrm{gr}_{\mathfrak{m}_A} A}.$$

Recall that a $\mathbb{C}$-algebra of finite type A is of the form $A = \mathbb{C}[x_1, \dots, x_n]/I$, for some $n \in \mathbb{N}$ and some ideal $I \subset \mathbb{C}[x_1, \dots, x_n]$. Equivalently, there is a closed immersion $\mathrm{Spec}(A) \hookrightarrow \mathbb{A}^n$ for some n.

Exercise 2.12. Compute h_R and show that if I is homogeneous, then

$$h_R = h_I + h_{R/I},$$

see Exercise 2.3.

Exercise 2.13. Let $I \subset R$ be a $\mathfrak{m}$-primary ideal. Then

- $h_{R/I}(0) = 1$
- $|h_{R/I}| = \sum_{i \geq 0} h_{R/I}(i) = \mathrm{len}\, R/I$.

Notice that the two together imply $0 \leq h_{R/I}(1) \leq (\mathrm{len}\, R/I) - 1$. Moreover, since R/I is a finite dimensional vector space, the function $h_{R/I}$ must vanish definitively. Therefore, we represent it as a vector, implicitly assuming that the values not displayed are zero.

Example 2.14. Consider the $\mathfrak{m}$-primary ideal $I = (x^2, xy, y^4)$. Then, as a vector space, R/I has the following direct sum decomposition

$$\mathbb{C}[x,y]/I = \underbrace{1 \cdot \mathbb{C}}_{(R/I)_0} \oplus \underbrace{x \cdot \mathbb{C} \oplus y \cdot \mathbb{C}}_{(R/I)_1} \oplus \underbrace{y^2 \cdot \mathbb{C}}_{(R/I)_2} \oplus \underbrace{y^3 \cdot \mathbb{C}}_{(R/I)_3}.$$

And we get

$$h_A = (1, 2, 1, 1).$$

Notice also that the $\mathbb{C}[x,y]$-module structure of the quotient $\mathbb{C}[x,y]/I$ is encoded in the diagram in Figure 3.

Precisely, moving right (resp. down) corresponds to the multiplication by x (resp. y).

1	x
y	
y^2	
y^3	

FIGURE 3. Graphical description of the R-module structure of R/I.

2.2.1. *Computation of the Hilbert–Samuel function.*

```
R = QQ[x,y]; mm=ideal(x,y);
I=ideal(x^2+y^3+5*y^4,y^5,x*y)
--The function tangentCone computes the initial ideal.
J=tangentCone I
-- M2 prints the value of the H--S function of R/J computed at i asking hilbertFunction(i,J)
for i from 0 to 10 list hilbertFunction (i,J)
for i from 0 to 10 list hilbertFunction (i,R/J)
for i from 0 to 10 list hilbertFunction (i,R)
```

Definition 2.15. The number $h_{R/I}(1)$ is called the embedding dimension of R/I. In symbols, we write $\mathrm{emb}_{R/I} = h_{R/I}(1)$.

Exercise 2.16. Let I be a $\mathfrak{m}$-primary ideal then $\operatorname{emb}_{R/I}$ is the smallest integer k such that there exists a closed immersion $\operatorname{Spec}(R/I) \hookrightarrow \mathbb{A}^k$ sending the unique closed point of the support of $\operatorname{Spec} R/I$ to the origin $0 \in \mathbb{A}^k$.

Hint: See Exercise 2.7 and Remark 2.8.

2.3. **Apolarity.** Apolarity (or Macaulay duality) is a powerful tool in commutative algebra. This technique has many application in algebraic geometry. For instance it helps in many cases to construct families of homogeneous ideals. The moral behind apolarity is:

construct the quotient instead of the ideal.

Working in characteristic 0 is crucial in this section.

Let us set

$$R = \mathbb{C}[x_1, \dots, x_n], \quad R^* = \mathbb{C}[y_1, \dots, y_n].$$

We view R^* as a R-module via the action

$$\begin{array}{ccc} R \times R^* & \longrightarrow & R^* \\ (x_1^{\alpha_1} \cdots x_n^{\alpha_n}, p(y_1, \dots, y_n)) & \longmapsto & \dfrac{\partial^{\sum_{i=1}^n \alpha_i}}{\partial^{\alpha_1} y_1 \cdots \partial^{\alpha_n} y_n} p, \end{array}$$

where $\alpha_i \in \mathbb{Z}_{\geq 0}$ for $i = 1, \dots, n$. This induces, for every $k \geq 0$, a perfect pairing

$$R_k \times R_k^* \longrightarrow R_0^* = \mathbb{C}$$

and, consequently, a notion of orthogonality.

We say that a vector subspace $T \subset R^*$ is graded if

$$T = \bigoplus_{k \in \mathbb{Z}} T \cap R_k^*.$$

Definition 2.17. An *inverse system* is a graded vector subspace $T \subset R^*$ closed under differentiation. If $S \subset R^*$ is a finite subset containing only homogeneous elements, then the inverse system generated by S is the smallest graded subspace $\langle S \rangle \subset R^*$ containing S and closed under differentiation. The *apolar ideal* attached to T is

$$T^\perp = \{\, r \in R \mid r \cdot T = 0 \,\} \subset R.$$

If $I \subset R$ is a homogeneous ideal, its associated inverse system is

$$I^\perp = \{\, r^* \in R^* \mid I \cdot r^* = 0 \,\} \subset R^*.$$

Example 2.18. Consider the subset $S = \left\{\, y_2^3, y_1 \,\right\} \subset R^*$. Then we have

$$\langle S \rangle = \left\{\, 1, y_1, y_2, y_2^2, y_2^3 \,\right\}.$$

Notice[3] that, if $I \subset \mathbb{C}[x, y]$ is the ideal in Example 2.14, then the classes in $\mathbb{C}[x, y]/I$ of the elements in $\langle S \rangle$ provide, after an appropriate relabelling of the variables, a basis for $\mathbb{C}[x, y]/I$.

Then we compute the apolar ideal to $\langle S \rangle$ and we get

$$\langle S \rangle^\perp = (x_1^2, x_1 x_2, x_2^4) \subset \mathbb{C}[x_1, x_2].$$

In particular, we get $R^*/\langle S \rangle^\perp \cong \mathbb{C}[x, y]/I$.

[3] Keep in mind the moral.

```
--Check for previous example
R = QQ[x,y]
ideal(y^4 , x*y,x^2) == inverseSystem matrix {{y^3,x}}
```

Example 2.18 is a special instance of a more general behaviour that we explain in Remark 2.19.

Remark 2.19. Notice that if $I \subset R$ is a homogeneous ideal, then $I^{\perp} \subset R^*$ is a graded subspace closed under differentiation. Conversely, every graded vector subspace $T \subset R^*$ closed under differentiation is orthogonal to the homogeneous ideal $T^{\perp} \subset R$. Moreover, if $V \subset R^*_j$ is a vector subspace, then

$$\dim_{\mathbb{C}}(V^{\perp})_j = \dim_{\mathbb{C}} R^*_j/V,$$

which yields an isomorphism of graded vector spaces $R/I \cong I^{\perp}$, see [4, Sec. 2].

2.4. **Monomial ideals and Partitions.** From now on we focus on monomial ideals, i.e. ideals admitting a generating set consisting only of monomials. In particular these ideals are homogeneous.

Definition 2.20. Fix $n, d \in \mathbb{Z}_{\geqslant 0}$. An $(n-1)$*-dimensional partition of size* d is a collection of d points $\lambda = \{\mathbf{a}_1, \dots, \mathbf{a}_d\} \subset \mathbb{N}^n$ such that if $\mathbf{y} \in \mathbb{N}^n$ satisfies $\mathbf{y} \leq \mathbf{a}_i$ for some $i = 1, \dots, d$, then $\mathbf{y} \in \lambda$. We call $|\lambda| = d$ the *size* of λ and we denote by P^n_d the set of $(n-1)$-dimensional partitions of size d, and by p^n_d the cardinality $|\mathrm{P}^n_d|$.

Exercise 2.21. Prove that the following associations are bijections

$$\begin{array}{ccccc}
\{\,\mathfrak{m}\text{-primary monomial ideals}\,\} & \longleftrightarrow & \left\{\begin{array}{c}\text{Subsets of } \mathbb{N}^n \text{ closed by translations}\\ \text{with finite complement}\end{array}\right\} & \longleftrightarrow & \{\,\text{Partitions}\,\} \\
I & \xmapsto{\quad\varphi\quad} & \left\{(\alpha_1, \dots, \alpha_n) \,\middle|\, \prod_{i=1}^n x_i^{\alpha_i} \in I\right\} & & \\
 & & A & \xmapsto{\quad\psi\quad} & A^c.
\end{array}$$

Definition 2.22. Let $(A, \mathfrak{m}_A)$ be a local Artinian $\mathbb{C}$-algebra of finite type. The socle of A is

$$\mathrm{Soc}(A) = (0_A : \mathfrak{m}_A) = \{\, a \in A \mid \mathfrak{m}_A \cdot a = 0 \,\}.$$

Exercise 2.23. Let $I \subset R$ be a $\mathfrak{m}$-primary ideal. Prove the following properties of $\mathrm{Soc}(R/I)$.

- The socle $\mathrm{Soc}(R/I)$ is a R-submodule of R/I.
- If I is a homogeneous ideal, then the socle $\mathrm{Soc}(R/I)$ is a graded R-module.
- If I is a monomial ideal, then $\mathrm{Soc}(R/I)$ corresponds to $\max \psi(\varphi(I))$, where φ, ψ are defined in Exercise 2.21.

2.4.1. *Computation of the socle of* R/I*, for* $I \subset R$ *monomial ideal.*

```
R = QQ[x,y]; mm=ideal(x,y);
-- Function to compute the socle of R/I with M monomial ideal
socleMonomial = I-> for u in (trim sub(I:mm,R/I))_* list sub(u,R);
J=ideal(x^2,y^5,x*y)
SOCLE = socleMonomial J
K = inverseSystem matrix {SOCLE}
K==J                                        -- Check that double-perp is the identity
```

Definition 2.24. A subset $S \subset \mathbb{N}^n$ is admissible if every two elements $s_1, s_2 \in S$ are not comparable with respect to the poset structure of $\mathbb{N}^n$.

Exercise 2.25. Prove that any admissible subset $S \subset \mathbb{N}^n$ is finite.
Hint: Induction.

Exercise 2.26. There is a bijection

$$\{\mathfrak{m}\text{-primary monomial ideals}\} \longleftrightarrow \{S \subset \mathbb{N}^n \text{ admissible}\}.$$

Hint: See Exercise 2.23.

2.5. **Special case $n = 2$.** The case $n = 2$ is special since, as we show in what follows, all admissible sets in dimension 2 are totally ordered.

Definition 2.27. Given two points $(n_1, n_2), (m_1, m_2) \in \mathbb{N}^2$ not comparable with respect to the poset structure of $\mathbb{N}^2$ we say that $(n_1, n_2) \prec (m_1, m_2)$ if

$$\begin{cases} m_2 > n_2, \\ n_1 > m_1. \end{cases}$$

Exercise 2.28. Definition 2.27 induces a total order on every admissible subset of $\mathbb{N}^2$.

Exercise 2.29. There is a bijection

$$\{\mathfrak{m}\text{-primary monomial ideals in 2 variables}\} \longleftrightarrow \{\text{totally } (\prec)-\text{ordered subsets of } \mathbb{N}^2\}.$$

Exercise 2.30. List all the monomial ideals in two variables of colength smaller or equal to 6.

Exercise 2.31. List all the pairs (I_1, I_2) of monomial ideals in two variables of respective colengths $d_1 \leq d_2 \leq 6$ such that $\psi(\varphi(I_1)) \subset \psi(\varphi(I_2))$ and depict the respective partitions.

2.5.1. *Pictorial description of the bijiection in Exercise 2.29.*

```
R = QQ[x,y]; mm=ideal(x,y); d=7;
--Function to depicts 1-partitions
drawPartition = d -> for a in d do (  print (for b from 0 to  a -1 list "*");)
--Function to convert partitions to ideals
partToideal = a -> trim(ideal(x^#a)+ideal(for b from 0 to #a-1 list y^(a#b)*x^b));
--Function to compute socle
socleMonomial = I->  for u in  (trim sub(I:mm,R/I))_* list  sub(u,R);
-- Computation of monomial ideals of colength d in 2 variables
P = partitions(d)
M= for a in P list partToideal a;
for p in P do (
    drawPartition p;
    print(" ");
    print(partToideal p);
    print("socle: ", socleMonomial partToideal p);
    print(" "); print(" ");
)
```

2.6. **The computation of** $\mathrm{Hom}_R(I, R/I)$ **for a monomial ideal** I**.** Let us recall the definition of minimal free resolution of a finitely generated R-module M.

Definition 2.32. Let M be a finitely generated R-module. A minimal free resolution is an exact sequence of the form

$$0 \longleftarrow M \longleftarrow F^\bullet, \tag{2.1}$$

where

$$F^\bullet: \qquad \cdots \longleftarrow F_{i-1} \xleftarrow{\delta_i} F_i \longleftarrow \cdots,$$

with all the F_i's free modules, i.e. $F_i \cong R^{n_i}$, for some $n_i \in \mathbb{N}$, and such that $\delta_i(F_i) \subset \mathfrak{m}F_{i-1}$.

Remark 2.33. The elements in F_0 are then called generators of M. The name generators is related to the fact that the first map $F_0 \xrightarrow{\delta_0} M$ in (2.1) is surjective, thus the image of a generating set for F_0 via δ_0 is a generating set for M.

The elements in F_1 are called first syzygies. This term means relation. Since the sequence (2.1) is exact, we have $\ker\delta_0 = \operatorname{Im}\delta_1$, i.e. any element in F_1 gives a relation between the generators of M, via the map δ_1.

The following theorem due to Hilbert guarantees that, when working in the polynomial setting, minimal free resolutions have finite length.

Proposition 2.34 ([1, Hilbert syzygy theorem])**.** *Any graded $\mathbb{C}[x_1,\dots,x_n]$-module M of finite type has a graded free resolution of length at most n.*

Exercise 2.35. Compute a minimal free resolution of the R module R/I where I is the ideal of the twisted cubic, see Definition 1.5.

2.6.1. *The minimal free graded resolution of the ideal of twisted cubic.*

```
n=3; R = QQ [x_0 .. x_n];
-- Computation of the ideal I_C of the twisted cubic
M = matrix { apply(n,i->R_*#i),apply(n,i->R_*#(i+1)) }
C = trim minors(2,M);
D = res C                                   -- res C computes the resolution of R/C
for i from 1 to n-1 do (
    print D.dd_i;
    print(" ");
)                                           -- Differential of the resolution
D.dd_1*D.dd_2                               -- Check it is a complex
```

Given an ideal $I \subset R$, the space of homomorphisms $\operatorname{Hom}_R(I,R/I)$ provides very important geometrical informations about the zero locus of I. Precisely it encodes the infinitesimal first order deformations of $\operatorname{Spec}(R/I)$ as we will see in Lecture 4.

We clearly have an injection

$$\operatorname{Hom}_R(I,R/I) \xhookrightarrow{\iota} \operatorname{Hom}_{\mathbb{C}}(I,R/I).$$

On the other hand a $\mathbb{C}$-linear homomorphism $\varphi \in \operatorname{Hom}_{\mathbb{C}}(I,R/I)$ is also R-linear if and only if it is compatible with the syzygies of I, i.e. given a relation

$$\sum_{j=1}^{s} p_j i_j = 0 \in R,$$

for some $p_j \in R$, $i_j \in I$, for $j = 1,\dots,s$, we require

$$\sum_{j=1}^{s} p_j \varphi(i_j) = 0 \in R/I.$$

Remark 2.36. These are the basics of deformation theory, see Lectures 3 and 4.

Exercise 2.37. If $I \subset R$ is a monomial ideal and $\{ m_1, \dots, m_s \}$ is a minimal set of monomial generators, then any R-linear homomorhism is uniquely determined by the images of the m_i's.

Exercise 2.38. Let $I \subset R$ be a $\mathfrak{m}$-primary homogeneous ideal.

- Show that partial derivatives $\frac{\partial}{\partial x_i}$ are well defined and linearly independent elements of $\mathrm{Hom}_R(I, R/I)_{-1}$.
- Show that any $\mathbb{C}$-linear homomorphism in $\mathrm{Hom}_{\mathbb{C}}(I, R/I)$ with target in the socle $\mathrm{Soc}(R/I)$ is naturally a R-homomorphism.

The following theorem is one possible incarnation of the celebrated result by Fogarty [7] about smoothness of the Hilbert scheme of points on a smooth surface.

Theorem 2.39. *Let $I \subset R = \mathbb{C}[x, y]$ be a $\mathfrak{m}$-primary homogeneous ideal. Then we have*

$$\dim_{\mathbb{C}} \mathrm{Hom}_R(I, R/I) = 2 \dim_{\mathbb{C}} R/I.$$

On the other hand the dimension of the non-negative part of the space $\mathrm{Hom}_{\mathbb{C}[x,y]}(I, \mathbb{C}[x, y]/I)$ was computed independently by Iarrobino and Briançon in [11] and [3] respectively.

Theorem 2.40 ([3, Thm. III.3.1] and [11, Thm 1])**.** *Let $I \subset R = \mathbb{C}[x, y]$ be a $\mathfrak{m}$-primary homogeneous ideal and let $h_{R/I}$ be the associated Hilbert–Samuel function. Denote[4] by $d, s > 0$ the integers such that*

$$h_{R/I} = (1, 2, \dots, d, h_d, \dots, h_{d+s-1}, 0, \dots),$$

with $h_d < d + 1$ and $h_i \geqslant h_{i+1}$, for $i \geqslant d$. Then, we have

$$\dim_{\mathbb{C}} \bigoplus_{k\geq 0} \mathrm{Hom}_R(I, R/I)_k = |h_{R/I}| - d - \sum_{i \geqslant d} \binom{h_{i-1} - h_i}{2}, \tag{2.2}$$

and

$$\dim_{\mathbb{C}} \mathrm{Hom}_R(I, R/I)_0 = \sum_{i \geqslant d} (h_{i-1} - h_i + 1)(h_i - h_{i+1}).$$

Example 2.41. Consider the ideal $I = \mathfrak{m}^2 = (x^2, xy, y^2)$ and let $\varphi \in \mathrm{Hom}_R(I, R/I)$ be a R-linear homomorphism. Since I is monomial, by Exercise 2.37, the homomorphism φ is uniquely determined by the images of x^2, xy, y^2.

Consider, for $A = (\alpha, \alpha_x, \alpha_y, \beta, \beta_x, \beta_y, \gamma, \gamma_x, \gamma_y) \in \mathbb{C}^6$ the $\mathbb{C}$-linear homomorphism

$$\begin{array}{ccc} \mathrm{Span}_{\mathbb{C}}(x^2, xy, y^2) & \xrightarrow{\overline{\varphi}_A} & R/I \\ x^2 & \longmapsto & \alpha + \alpha_x x + \alpha_y y \\ xy & \longmapsto & \beta + \beta_x x + \beta_y y \\ y^2 & \longmapsto & \gamma + \gamma_x x + \gamma_y y. \end{array}$$

Then, the map $\overline{\varphi}_A$ extends to a (unique) R-linear homomorphism $\varphi_A \in \mathrm{Hom}_R(I, R/I)$ if and only in the following conditions are satisfied

$$y\overline{\varphi}_A(x^2) - x\overline{\varphi}_A(xy) = 0, \quad x\overline{\varphi}_A(y^2) - y\overline{\varphi}_A(xy) = 0, \quad y^2\overline{\varphi}_A(x^2) - x^2\overline{\varphi}_A(y^2) = 0.$$

[4]Notice that there always exist such d, s.

This imposes the conditions

$$\alpha=\beta=\gamma=0. \tag{2.3}$$

Notice that, since the last syzygy is algebraically dependent from the first two, it is enough to consider the first two syzygies to compute the module $\mathrm{Hom}(I,R/I)$.

We also remark that this result can be obtained via a different argument. Indeed, any $\mathbb{C}$-homomorphism taking values in the socle is R-linear by Exercise 2.38. This, together with Fogarty's result, see Theorem 2.39, already implies conditions (2.3). As expected, we get

$$\dim_{\mathbb{C}} \mathrm{Hom}_R(I,R/I)=2\cdot 3=6.$$

Notice that the tangent space is concentrated in degree -1.

2.6.2. *Computations from Example 2.41.*

```
R = QQ[x,y];
m2 =  (ideal R_*)^2;                          -- Square of the maximal ideal
N = Hom(m2,R/m2)                              -- Computation of Hom module
hilbertFunction (-1,N)                        -- Degree -1 H--S function of N
F = res m2;                                   -- Resolution of R/m2
A = F.dd_1                                    -- Generators
B = F.dd_2                                    -- First syzygies
-- Straightforward computation of Hom(m2,R/m2)
S = R[a_1..a_3,b_1..b_3,c_1..c_3]
m2 = sub(m2,S);
Syz_1 = entries (B_0);
Syz_2 = entries (B_1);
-- Impose syzygies to the image of a morphism
Im = for i from 1 to 3 list (a_i + b_i *x + c_i*y)                      -- Image of a morphism
Check = ideal for j from 1 to 2 list sum for i from 0 to 2 list (Im#i)*(Syz_j#i) -- Imposition
sub(Check , S/m2)                                          -- Recover the conditions a_1=a_2=a_3=0
```

Exercise 2.42. Compute the modules $\mathrm{Hom}_R(I,R/I)$, for I one of the following ideals.

- $I=(x,y)\subset\mathbb{C}[x,y]$,
- $I=(x,y,z)^2\subset\mathbb{C}[x,y,z]$,
- $I=(x,y)^2+(z,t)^2+(xz-yt)\subset\mathbb{C}[x,y,z,t]$.

In the third case show that

$$\dim_{\mathbb{C}} \mathrm{Hom}_R(I,R/I)_{-1}=4.$$

In particular, by Exercise 2.38, we get

$$\mathrm{Hom}_R(I,R/I)_{-1}=\mathrm{Span}_{\mathbb{C}}\left(\frac{\partial}{\partial x},\frac{\partial}{\partial y},\frac{\partial}{\partial z},\frac{\partial}{\partial t}\right).$$

2.6.3. *Some computations for Exercise 2.42.*

```
-- Ideal I1
R1 = QQ[x,y]; I1 = ideal R1_*; N1 =Hom(I1,R1/I1);
-- Ideal I2
R2 = QQ[x,y,z]; I2 = (ideal R2_*)^2; N2 =Hom(I2,R2/I2);
-- Ideal I3
R3=QQ[x,y,z,t]; I3=(ideal(x,y))^2+(ideal(z,t))^2+det(matrix{{x,z},{y,t}}); N3=Hom(I3,R3/I3);
-- Dimension of tangent spaces
(degree N1, degree N2, degree N3)
```

```
-- Degree -1 part of N3
hilbertFunction(-1,N3)
```

Exercise 2.43. Consider the power of the maximal ideal $(x, y)^k \subset \mathbb{C}[x, y]$, for some $k \geq 1$. Then $\mathrm{Hom}_R(I, R/I)$ is concentrated in degree -1.

```
n= 3; k= 4; R = QQ[x_1..x_n]; m=ideal R_*;
N=Hom(m^k,R/m^k);
-- User-friendly output
for i from -2 to 2 do print("h("|toString i |")="|toString hilbertFunction(i,N))
```

3. Lecture 3

In this lecture we introduce the notion of a (flat) family of subschemes. We will first focus on families over the spectrum of dual numbers and then show how these families can be lifted to other Artinian $\mathbb{C}$-algebras of finite type. The theory will be presented mainly by means of working examples. The last part deals with the concept of limit of a flat family.

3.1. **Notation.**

- We denote by $\mathbb{A}^n_{\alpha_1,\ldots,\alpha_n}$ the affine space $\operatorname{Spec}\mathbb{C}[\alpha_1,\ldots,\alpha_n]$ to emphasise the choice of coordinates.
- Similarly, we write $\mathbb{P}^n_{x_0,\ldots,x_n}$ to underline the choice of homogeneous coordinates $x_0,\ldots,x_n$ on $\mathbb{P}^n$.
- Given two varieties X, Y, and two points $p \in X, q \in Y$ points, we denote by $(X, p) \xrightarrow{\varphi} (Y, q)$ a morphism from X to Y sending the point p to the point q.

3.2. **The spectrum of dual numbers.**

The easiest example of non reduced scheme is the double point.

Definition 3.1. The ring of dual numbers is the quotient ring $\mathbb{C}[t]/(t^2)$. According to the standard notation in the literature we denote it by $\mathbb{C}[\varepsilon]$, where ε is the class of t in the quotient.

In what follows we denote its spectrum by $D_\varepsilon = \operatorname{Spec}\mathbb{C}[\varepsilon]$.

Remark 3.2. The scheme D_ε has a unique closed point corresponding to the maximal ideal $(\varepsilon) \subset \mathbb{C}[\varepsilon]$. The big difference from the one-point scheme $\operatorname{Spec}\mathbb{C}$ is that the "ring of functions" of D_ε, i.e. its coordinate ring, does not only consist in constants but it contains classes of polynomials of degree at most one, see Remark 3.3 for more details. Indeed, we have an isomorphism of vector spaces

$$\mathbb{C}[\varepsilon] \cong_{\mathbb{C}} \mathrm{Span}_{\mathbb{C}}(1, \varepsilon).$$

Remark 3.3. The ring of dual numbers (or equivalently its spectrum) is the algebraic geometrical analogue of the notion of *"infinitesimal"* from other subjects. To give an idea, we present some informal explanations below.

- In analysis the small-O in Landau notation stands for an infinitesimal function around a given point $p \in \mathbb{R}$, i.e. a function whose powers are negligible with respect to constant functions.
- In differential geometry, partial derivatives describe the local properties of a function in an arbitrary small neighbourhood of a given point.

Another useful interpretation is to think of the spectrum of dual numbers as the scheme theoretic intersection of a conic and a line tangent to it. For example, it can be obtained as the limit of the intersection of a conic with a secant line that tends to be tangent to the conic.

Remark 3.4. In order to realise D_ε as a closed subscheme of the affine space $\mathbb{A}^n$ supported at the origin one has to fix a line $L = \operatorname{Spec}\mathbb{C}[x_1,\dots,x_n]/(\ell_1,\dots\ell_{n-1}) \subset \mathbb{A}^n$, where $\ell_i \in R_1$ is a linear form, for $i = 1,\dots,n$ and then intersect it with the closed subscheme of $\mathbb{A}^n$ defined by the vanishing of the square of maximal ideal $(x_1,\dots,x_n)^2$.

To make an example, suppose that L is defined by the vanishing of the linear forms $x_1,\dots,x_{n-1}$, then we get the isomorphism

$$\frac{\mathbb{C}[x_1,\dots,x_n]}{(x_1,\dots,x_{n-1})+\mathfrak{m}^2} = \frac{\mathbb{C}[x_1,\dots,x_n]}{(x_1,\dots,x_{n-1},x_n^2)} \cong \frac{\mathbb{C}[x_n]}{(x_n^2)} \cong \mathbb{C}[\varepsilon].$$

Usually, D_ε is depicted as a closed point, together with a vector corresponding to the direction of L.

FIGURE 4. Pictorial description of D_ε.

Exercise 3.5. Let Z be an affine scheme of length 2 with a unique closed point. Then, Z is isomorphic (as $\mathbb{C}$-scheme) to the spectrum of the ring of dual numbers, i.e.

$$Z \cong \operatorname{Spec}\mathbb{C}[\varepsilon].$$

Hint: Fix the 2-dimensional vector space $V \cong \mathbb{C}^2$, a polynomial ring $R = \mathbb{C}[x_1,\dots,x_n]$ with complex coefficients and the maximal ideal $\mathfrak{m} \subset R$ generated by the variables. You need to endow V of a structure of R-module in such a way there is a vector $v \in V$ such that $R \cdot v = V$. This is the same as to fix linear maps $X_i \in \mathrm{GL}(V)$, for $i = 1,\dots,n$, with suitable properties:

- the X_i's commute,
- the R-module structure induced on V is annihilated by $\mathfrak{m}$,
- the vector $v \in V$ generates V via the R-action, i.e. $\mathrm{Span}(X_i \cdot v \mid i = 1,\dots,n) = V$.

For instance, if $R = \mathbb{C}[x,y]$ we obtain

$$X = \begin{pmatrix} 0 & 0 \\ \alpha & 0 \end{pmatrix} \quad Y = \begin{pmatrix} 0 & 0 \\ \beta & 0 \end{pmatrix}.$$

And the R-module structure induced on V is

$$V \cong \frac{\mathbb{C}[x,y]}{(\alpha y - \beta x)+\mathfrak{m}^2}.$$

3.2.1. *Solution to exercise Exercise 3.5.*

```
R=QQ[a_(1,1)..a_(2,2),b_(1,1)..b_(2,2)];
X= sub (transpose genericMatrix (R,a_(1,1),2,2),{a_(1,2)=>0} );
Y= sub(transpose genericMatrix (R,b_(1,1),2,2),{b_(1,2)=>0} );
Cond1 = radical trim ideal (flatten entries X^2)                -- Nilpotence
Cond2 = radical trim ideal (flatten entries Y^2)                -- Nilpotence
Cond3 = radical trim ideal (flatten entries X*Y)                -- Nilpotence
Cond4 = radical trim ideal (flatten entries (X*Y-Y*X))          -- Commutativity
Cond = trim ( Cond1 + Cond2 + Cond3 + Cond4 )
X = sub( sub (X, R/Cond) , R)
Y = sub( sub (Y, R/Cond ) , R)
```

Exercise 3.6. Show that, if $Z \subset \mathbb{A}^n$ is any zero-dimensional closed subscheme of length 2, then there exists a unique line $L \subset \mathbb{A}^n$ such that $Z \subset L$.

3.3. **The tangent space in algebraic geometry.** Let X be a variety and let $p \in X$ be a closed point. The tangent space T_pX has different characterisations. We recall now the most classical definition and then we discuss an equivalent formulation.

Definition 3.7. Let X be a variety and let $p \in X$ be a closed point. The tangent space T_pX is the following vector space

$$T_pX = (\mathfrak{m}_p/\mathfrak{m}_p^2)^\vee,$$

where $\mathfrak{m}_p \subset \mathscr{O}_{X,p}$ is the maximal ideal of the local ring $\mathscr{O}_{X,p}$.

As in differential geometry, where the tangent space is identified with the space of derivatives, in algebraic geometry there is a similar interpretation in terms of (pointed) closed immersions $(D_\varepsilon,(\varepsilon)) \to (X,p)$, see Remark 3.3.

Exercise 3.8. Let X be a variety and let $p \in X$ be a closed point. The tangent space T_pX is isomorphic to the following vector space

$$T_pX \cong \left\{ \varphi : (D_\varepsilon,(\varepsilon)) \to (X,p) \right\}.$$

Notice that the exercise also requires to endow the left hand side of a structure of complex vector space.

Hint: We need a local ring homomorphism $\mathscr{O}_{X,p} \xrightarrow{\varphi} \mathbb{C}[\varepsilon]$ sending the unique maximal ideal $\mathfrak{m}_p \subset \mathscr{O}_{X,p}$ to $(\varepsilon) \subset \mathbb{C}[\varepsilon]$. Start showing that $\mathfrak{m}_p^2$ must be contained in $\ker\varphi$.

Example 3.9. In this example we compute the tangent space to the singular point of the nodal cubic.

Let $C \subset \mathbb{A}^2$ be the curve defined by the vanishing of the polynomial $f = y^2 - x^2(x+1)$. Then, the Jacobian criterion tells us that the unique singular point of C is the origin $(0,0) \in C \subset \mathbb{A}^2$. Indeed, the Jacobian of f is

$$\operatorname{Jac} f = \begin{pmatrix} -3x^2 - 2x \\ 2y \end{pmatrix}$$

and its rank drops, on C, only for $x = y = 0$.



FIGURE 5. Pictorial description of two tangents to the nodal cubic at the origin.

Let us compute the tangent space to C at the origin in terms of Exercise 3.8. A morphism $D_\varepsilon \to C$ sending the unique closed point of D_ε to the origin, corresponds to a ring homomorphism

$$\frac{\mathbb{C}[x,y]}{(f)} \xrightarrow{\varphi^\#} \mathbb{C}[\varepsilon]$$

sending the maximal[5] ideal $(x,y) \subset \mathbb{C}[x,y]/(f)$ to (ε). This is equivalent to endow the ring $\mathbb{C}[\varepsilon]$ of a structure of $\mathbb{C}[x,y]$-module, see Remark 3.4.

[5]We use the same notation for the variables and their classes in the quotient ring. Be aware of this fact.

In other words we ask for which $[a:b]\in\mathbb{P}^1$ the association

$$\begin{array}{ccc} \frac{\mathbb{C}[x,y]}{(f)} & \xrightarrow{\varphi^\#} & \frac{\mathbb{C}[x,y]}{(ax+by)+\mathfrak{m}^2} \\ 1 & \longmapsto & 1, \end{array}$$

is a well defined $\mathbb{C}[x,y]$-linear homomorphism. This is the case if and only if $f\in(ax+by)+\mathfrak{m}^2$.

Denote by $I_{[a:b]}$ the ideal $I_{[a:b]}=(ax+by)+\mathfrak{m}^2$. If we put $b=0$, then we clearly have

$$f\in I_{[1:0]}=(x,y^2).$$

If we put $b\neq 0$ then we have

$$f=(y+ax)(y-ax)+x^2(a-1-x)\in I_{[a:1]}.$$

Therefore, we recover the tangent space to the nodal cubic at its singular point as the tangent space of the whole affine plane at the origin $(0,0)\in\mathbb{A}^2$.

3.3.1. *Computations with the nodal cubic.*

```
R=QQ[x,y]; f = y^2 - x^2*(x+1);
Sings = trim (ideal(f)+ minors(1, jacobian f));       -- Jacobian ideal
Sings = trim oo                                       -- Minimal generators
-- Ambient variables for the pencil of lines trough the (0,0)
S = R[a,b];
f = sub (f, S);
I= trim ( ideal(a*x+b*y) + (ideal(x,y))^2)            -- Generic first order vector at (0,0)
I + f == I                                            -- Check f is in I
```

Exercise 3.10. The quadric cone is the hypersuperface $Q=\left\{x^2-yz=0\right\}\subset\mathbb{A}^3$.

- Show that the unique singular point of Q is the origin $(0,0,0)\in\mathbb{A}^3$.
- Compute the tangent space in terms of closed immersions $D_\varepsilon\hookrightarrow Q$.
- Show that the quadric cone is a cone with vertex at the origin, i.e. that for any $p\in Q$ the line $L_p=\langle(0,0,0),p\rangle$ generated by p, and the origin is entirely contained in Q.

Consider the exact sequence

$$0\longrightarrow (t^2)/(t^3)\longrightarrow \mathbb{C}[t]/(t^3)\xrightarrow{\pi}\mathbb{C}[t]/(t^2)\longrightarrow 0.$$

It is natural to ask if a tangent vector $\varphi:D_\varepsilon\to X\subset\mathbb{A}^n$ lifts to a morphism $\operatorname{Spec}\mathbb{C}[t]/t^3\xrightarrow{\overline{\varphi}}X$, i.e. if there exists a morphism $\overline{\varphi}$ making the following diagram commutative

$$\begin{tikzcd} \mathbb{C}[x_1,\dots,x_n]/I_X \arrow[r, "\varphi"] \arrow[dr, "\overline{\varphi}"'] & \mathbb{C}[t]/(t^2) \\ & \mathbb{C}[t]/(t^3), \arrow[u, "\pi"'] \end{tikzcd}$$

where $I_X\subset\mathbb{C}[x_1,\dots,x_n]$ denotes the ideal of $X\subset\mathbb{A}^n$.

Remark 3.11. Notice that the existence of the lifting is a local problem, and it depends only on the kind of singularity (X,p).

In particular, the hypothesis $X\subset\mathbb{A}^n$ is not restrictive as our problem is local in nature and any point of a scheme has an affine neighbourhood.

Example 3.12. Let us consider again the nodal cubic $C \subset \mathbb{A}^2$, see Example 3.9. In this example we detect tangents to C at the origin that can be lifted to $\mathbb{C}[t]/(t^3)$. Non-explicit computations are given in the M2 code below.

In practice, we ask for which polynomial

$$ax + by + cx^2 + dxy + ey^2 \in \mathbb{C}[x, y],$$

it is possible to build a commutative diagram of the form

$$\begin{tikzcd} \frac{\mathbb{C}[x,y]}{f} \arrow[r] \arrow[dr] & \frac{\mathbb{C}[x,y]}{(ax+by)+\mathfrak{m}^2} \arrow[d] \\ & \frac{\mathbb{C}[x,y]}{(ax+by+cx^2+dxy+ey^2)+\mathfrak{m}^3}. \end{tikzcd} \tag{3.1}$$

First, we put $b = 0$ and thus $a \neq 0$, and we ask

$$f \overset{?}{\subset} (ax + cx^2 + dxy + ey^2) + \mathfrak{m}^3.$$

This question has clearly a negative answer, see M2 computations below if you need further confirmation. On the other hand, if we put $b \neq 0$, a direct computation shows that $f \subset (y + ax + cx^2 + dxy + ey^2) + \mathfrak{m}^3$ if and only if

$$a^2 = b^2 \neq 0,$$

where the last inequality is consequence of the commutativity of the diagram (3.1), and of Example 3.9 where we required $[a : b] \in \mathbb{P}^1$.

3.3.2. *Computations for Example 3.12.*

```
U = frac(QQ[a,b,c,d,e]); T = U[x,y];
f = y^2 - x^2*(x+1);
curv = a*x+b*y+c*x^2+d*x*y+e*y^2 ;
I= trim ( ideal(curv) + (ideal(x,y))^3) ;
I + f == I
C = coefficients sub(sub(f,T/I),T);
Sol = ideal for u in apply (flatten entries C#1,d-> sub(d,U)) list numerator u;
minimalPrimes Sol
J1 = sub(I , {a=>1,b=>1}) ;
J2 = sub(I , {a=>1,b=>-1}) ;
J3 = sub(I , {a=>1,b=>-2}) ;
J1 + f == J1
J2 + f == J2
J3 + f == J3
```

3.4. **Families of ideals of finite colength.**

Definition 3.13. Let X be a quasi-projective variety and let B be a scheme over $\mathbb{C}$. A family of closed subschemes of X parametrised by B (or a B-family of closed subschemes of X) is a closed B-flat subscheme

$$Z \subset B \times X.$$

When $B \cong \mathbb{C}[x_1, \ldots, x_n]/I$ with $\sqrt{I}$ maximal, we say that Z is an infinitesimal family.

When $B = D_\varepsilon$ is the spectrum of dual numbers, we say that Z is a first order (infinitesimal) family[6], see Lecture 4.

Flatness is a classical algebraic notion, see [6, §I.6]. Since we are interested on families of zero-dimensional schemes, or families parametrised by the spectrum of dual numbers D_ε, we report two sufficient conditions for flatness in these settings.

Theorem 3.14 ([10, Chap. III Theorem 9.9])**.** *Let B, X be two quasi projective varieties with B integral. Let also $Z \subset B \times X$ be a closed subscheme. Suppose also that for any $b \in B$ the fibre $Z_b = Z \cap (\{b\} \times X)$ is zero dimensional. Then, Z is B-flat if and only if the association*

$$b \mapsto \operatorname{len} Z_b$$

is constant.

The second criterion for flatness we present is known as Artin criterion and it expresses flatness over $\operatorname{Spec} A$, for A local Artinian algebra, in terms of syzygies of the so-called *"central fibre"*, i.e. the fibre over the unique closed point of $\operatorname{Spec} A$, we will expand on this in Lecture 4.

Theorem 3.15 (Artin criterion for flatness [2])**.** *Let $R = \mathbb{C}[x_1, \ldots, x_n]$ be the polynomial ring in n variables and complex coefficients. Let also A be a local Artinian algebra of finite type over $\mathbb{C}$. Let also $I \subset R$ and $I_A \subset R_A = R \otimes A$ be two ideals. Suppose that there is an exact sequence*

$$R^\ell \to R^m \to R \to R/I \to 0 \tag{3.2}$$

and a complex

$$R_A^\ell \to R_A^m \to R_A \to R_A/I_A \to 0 \tag{3.3}$$

such that the part

$$R_A^m \to R_A \to R_A/I_A \to 0$$

is exact. Then, if (3.3) $\otimes_A \mathbb{C} =$ (3.2)*, the module R_A/I_A is A-flat.*

Remark 3.16. Notice that Artin criterion is precisely what we asked in Lecture 2.6 to a $\mathbb{C}$-linear homomorphism in order to be R-linear. This is not a case as we will show in Lecture 4.

Exercise 3.17. Prove the following facts about infinitesimal families.

- Show that the *"graph"* of a tangent vector defines a first order family, i.e. show that the fibre product

$$\begin{tikzcd} D_\varepsilon \times_X X \arrow[r] \arrow[d] & X \arrow[d, "\mathrm{id}_X"] \\ D_\varepsilon \arrow[r] & X \end{tikzcd}$$

  defines a first order family $D_\varepsilon \times_X X \subset D_\varepsilon \times X$.
- Interpret the lifts in Example 3.12 as an infinitesimal family.

Exercise 3.18. This exercise suggests the relation between (lifts to spectra of local Artinian algebras of) tangent vectors and (infinitesimal) first order families.

[6]This terminology is not classical, in Lecture 4 we introduce the notion of infinitesimal deformation solving this notational problem.

- In the same setting of Example 3.12, find conditions on the complex numbers u, v, r, s such that the infinitesimal family corresponding to the ideal $(y+x+cx^2+dxy+ey^2)+\mathfrak{m}^3$, where $c, d, e \in \mathbb{C}$, lifts to

$$\frac{\mathbb{C}[x,y]}{(y+x+cx^2+dxy+ey^2+ux^3+vx^2y+rxy^2+sy^3)+\mathfrak{m}^4}.$$

- Show that a tangent vector $(D_\varepsilon,(\varepsilon)) \hookrightarrow (Q,(0,0,0))$ lifts to an infinitesimal family

$$(\operatorname{Spec}\mathbb{C}[t]/(t^3),(t)) \hookrightarrow (Q,(0,0,0)),$$

if and only if it factors through as shown in the following commutative diagram

$$\begin{tikzcd}
(\operatorname{Spec}\mathbb{C}[t]/(t^3),(t)) \arrow[r, hook] \arrow[dr, hook] & (Q,(0,0,0)) \\
& (\mathbb{A}^1,0). \arrow[u, hook]
\end{tikzcd}$$

3.4.1. *Computations for Exercise 3.18.*

```
V = frac(QQ[c,d,e,u,v,r,s]);
O = V[x,y]
f = y^2 - x^2*(x+1);
curv = x+y+c*x^2+d*x*y+e*y^2+u*x^3 +v*x^2*y + r*x*y^2+s*y^3
I= trim ( ideal(curv ) + (ideal(x,y))^4)
I + f == I
C = coefficients sub(sub(f,O/I),O);
Sol = ideal for k in apply (flatten entries C#1,d-> sub(d,V)) list numerator k;
Sol = (minimalPrimes Sol )#0_*#0
J = sub( I ,  c=> -(- 2*d + 2*e - 1)/2)
J + f == J
```

Example 3.19. We put $X = \mathbb{A}^1_x$ and $B = \mathbb{A}^1_\alpha$. Then, a family of closed subschemes of X parametrised by B is a closed subscheme $Z \subset \mathbb{A}^1_x \times \mathbb{A}^1_\alpha \cong \mathbb{A}^2_{x,\alpha} = \operatorname{Spec}\mathbb{C}[x,\alpha]$ flat over $\mathbb{A}^1_\alpha$. Therefore, to define Z, we need to give an ideal $I_Z \subset \mathbb{C}[x,\alpha]$ defining an $\mathbb{A}^1_\alpha$-flat family Z, i.e. we look for a commutative diagram of the following form,

$$\begin{tikzcd}
Z \arrow[r, hook] \arrow[dr] & \mathbb{A}^1_\alpha \times X \arrow[d] \\
& \mathbb{A}^1_\alpha,
\end{tikzcd}$$

where Z is $\mathbb{A}^1_\alpha$-flat.

Let us pick the ideal $I_Z = (x-\alpha) \subset \mathbb{C}[x,\alpha]$, defining the diagonal $\Delta \subset \mathbb{A}^2_{x,\alpha}$. Notice that, for all $\alpha_0 \in \mathbb{C}$, we have a commutative diagram

$$\begin{tikzcd}
\frac{\mathbb{C}[x,\alpha]}{(\alpha-\alpha_0,x-\alpha)} \cong \frac{\mathbb{C}[x]}{(x-\alpha_0)} & \frac{\mathbb{C}[x,\alpha]}{(x-\alpha)} \arrow[l, two heads] & \mathbb{C}[x,\alpha] \arrow[l, two heads] \\
\frac{\mathbb{C}[\alpha]}{(\alpha-\alpha_0)} \arrow[u, hook] & & \mathbb{C}[\alpha], \arrow[ll, two heads] \arrow[u, hook] \arrow[ul, hook]
\end{tikzcd}$$

which guarantees flatness of the family as $\operatorname{len} Z_{\alpha_0} = 1$ for all $\alpha_0 \in \mathbb{C}$, see Theorem 3.14. Figure 6 is a pictorial description of the family Z.



FIGURE 6. Pictorial description of the family Z. The family is the Yellow curve and the fibre over α_0 is the red point.

Example 3.20. Again, as in Example 3.19 we put $X = \mathbb{A}^1_x$ and $B = \mathbb{A}^1_\alpha$. This time, we consider the ideal $I_Z = (x-1)(x-\alpha) \subset \mathbb{C}[x,\alpha]$ defining the union of the diagonal $\Delta \subset \mathbb{A}^2_{x,\alpha}$ and the line having equation $x = 1$.

When $\alpha_0 \in \mathbb{C} \setminus 1$, the fibre Z_{α_0} consists of two distinct points, see Figure 7.



FIGURE 7. The fibre over $\alpha_0 \neq 1$ consists of the two red points of $X = \mathbb{A}^1_x$.

Now we look at the fibre over the point $1 \in \mathbb{A}^1_\alpha$. This is computed via the following commutative diagram

$$\begin{tikzcd}
\frac{\mathbb{C}[x,\alpha]}{(\alpha-1,(x-1)^2)} \cong \mathbb{C}[\varepsilon] & \frac{\mathbb{C}[x,\alpha]}{(x-1)(x-\alpha)} \arrow[l, two heads] & \mathbb{C}[x,\alpha] \arrow[l, two heads] \\
\frac{\mathbb{C}[\alpha]}{(\alpha-1)} \arrow[u, hook] & & \mathbb{C}[\alpha], \arrow[ll, two heads] \arrow[u, hook] \arrow[ul, hook]
\end{tikzcd}$$

whose geometric counterpart is

$$\begin{tikzcd}
D_\varepsilon \arrow[r, hook] \arrow[d, two heads] & Z \arrow[r, hook] \arrow[dr] & \mathbb{A}^2_{x,\alpha} \arrow[d, two heads] \\
\{1\} \arrow[rr, hook] & & \mathbb{A}^1_\alpha.
\end{tikzcd} \tag{3.4}$$

Summarising, all the fibres have length 2 and thus Z is flat over $\mathbb{A}^1_\alpha$ and it defines an $\mathbb{A}^1_\alpha$-family of closed subschemes of X, see Theorem 3.14.

Again, in Figure 8 we present a pictorial version of the diagram Equation (3.4).



FIGURE 8. Pictorial counterpart of the commutative diagram Equation (3.4).

The following exercise explains how to build a D_ε-family starting from a morphism $\varphi \in \mathrm{Hom}_R(I, R/I)$, see [2] for more details.

Exercise 3.21. Let $I \subset R$ be a $\mathfrak{m}$-primary ideal and let $\varphi \in \mathrm{Hom}_R(I, R/I)$ be a R-linear homomorphism. Then, the ideal

$$(f + \varepsilon \cdot \widetilde{\varphi(f)} \mid f \in I, \text{ and } \widetilde{\varphi(f)} \in R \text{ is a lifting of } \varphi(f) \in R/I) \subset R \otimes \mathbb{C}[\varepsilon],$$

defines a D_ε-family of closed subschemes of $\mathrm{Spec}(R)$.

3.5. **Limits.** We conclude this lecture by discussing limits of (families of) ideals.

Theorem 3.22 ([10, Proposition 9.8])**.** *Let B be a smooth irreducible variety of dimension 1, and let $p \in B$ be a closed point. Let also $Z \subset (B \setminus p) \times \mathbb{P}^n$ be a closed subscheme flat over $B \setminus p$. Then, there exists a unique closed subscheme $\overline{Z} \subset B \times \mathbb{P}^n$ flat over B, whose restriction to $(B \setminus p) \times \mathbb{P}^n$ is Z.*

Remark 3.23. In Theorem 3.22 we can replace $\mathbb{P}^n$ with any X projective.

If the fibres of the restriction of the projection $\pi_B|_Z : Z \to B$ are zero-dimensional and all have the same support, we can replace $\mathbb{P}^n$ by $\mathbb{A}^n$. This is true because zero-dimensional schemes are the only projective schemes that can be embedded in affine spaces, and the requirement that they have the same support guarantees that they cannot escape at infinity.

Example 3.24. In order to deal with limits, it is crucial to work with $\dim Y = 1$.

Consider the ideal $I_{s,t} = (x, y) \cap (x - s, y - t) \subset \mathbb{C}[x, y] \otimes [s, t]_{(s,t)}$ defining a closed subscheme Z of the product $\mathbb{A}^2_{x,y} \times (\mathbb{A}^2_{s,t} \setminus (0,0)$. Then, we have

$$I = (x, y) \cap (x - s, y - t) = (x, y) \cdot (x - s, y - t),$$

where the last equality is a consequence of the fact that the ideals (x, y) and $(x - s, y - t)$ define disjoint subschemes of $\mathbb{A}^2_{x,y} \times (\mathbb{A}^2_{s,t} \setminus (0,0)$. Then, for any $(s_0, t_0) \in \mathbb{A}^2_{s,t} \setminus (0,0)$ the fibre $Z_{(s_0,t_0)}$ has length 2. On the other hand, the limit at the origin $(0,0) \in \mathbb{A}^2_{s,t}$ is

$$Z_{(0,0)} = \mathrm{Spec}\, R/(x, y)^2,$$

and it has length 3. Therefore, the closure subscheme $\overline{Z} \subset \mathbb{A}^2_{x,y} \times \mathbb{A}^2_{s,t}$ is not $\mathbb{A}^2_{s,t}$-flat and it does not define an $\mathbb{A}^2_{s,t}$-family of closed subschemes of $\mathbb{A}^2_{x,y}$, see Theorem 3.14.

See also Exercise 3.29

Remark 3.25. Naive limits do not work!!!

Let $Z \subset \mathbb{C}^* \times \mathbb{A}^n$ be a $\mathbb{C}^*$-family of zero-dimensional subschemes of $\mathbb{A}^n$, and let $I_Z \subset \mathbb{C}[s, s^{-1}] \otimes R$ be its defining ideal.

Suppose that we want to compute the ideal of the limit $\overline{Z}_0 \subset \mathbb{C}[s] \otimes R$ following Theorem 3.22. It is in general not enough to put $s = 0$ to get the required result. The correct procedure is to first saturate the ideal $I_Z \cap \mathbb{C}[s] \otimes R$ by the ideal (s) and then make the substitution $s \to 0$.

Geometrically, saturation corresponds to flattification, i.e. it corresponds to substitute the fibre over the origin of $I_Z \cap \mathbb{C}[s]$ with the unique fibre making the family flat, see Remark 3.27

Definition 3.26. Let $I, J \subset R$ be two ideals, then the saturation of I with respect to J is the following ideal

$$(I : J^\infty) = \bigcup_{d=1}^{\infty} (I : J^d).$$

Remark 3.27. Geometrically the ideal $(I:J)$ correspond to the following operation

$$\overline{\operatorname{Spec}(R/I)\setminus\operatorname{Spec}(R/J)}\subset\operatorname{Spec}(R).$$

Similarly, saturation remove from the zero-locus of I all the thickenings of the zero locus of J and then takes the closure.

Example 3.28. Let us see saturation in action.

Consider the ideal $I_s=(x,y)\cdot(x-s,y-s)\subset\mathbb{C}[x,y,s]$ defining a closed subscheme Z of the product $\mathbb{A}^2_{x,y}\times\mathbb{A}^1_s$. Then, for any $s_0\in\mathbb{A}^1_s\setminus 0$ the fibre Z_{s_0} has length 2. On the other hand

$$Z_0=\operatorname{Spec}R/(x,y)^2,$$

has length 3. Therefore, the subscheme $Z\subset\mathbb{A}^2_{x,y}\times\mathbb{A}^2_s$ is not $\mathbb{A}^2_s$-flat and it does not define an $\mathbb{A}^2_s$-family of closed subschemes of $\mathbb{A}^2_{x,y}$, see Theorem 3.14.

Now, we saturate the ideal I_s with respect to the ideal s and we get the ideal

$$\begin{aligned}\left(I_s:(s)^\infty\right)&=\left((x,y)\cdot(x-s,y-s),(s)^\infty\right)\\&=\left((x^2-xs,y^2-ys,xy-xs,xs-ys),(s)^\infty\right)\\&=(x^2-xs,x-y)\end{aligned}$$

defining a closed subscheme $\widetilde{Z}_s\subset\mathbb{A}^2_{x,y}\times\mathbb{A}^1_s$.

Finally, note that $Z_{s_0}=\widetilde{Z}_{s_0}$ for $s_0\in\mathbb{A}^1_s\setminus 0$, but now $\operatorname{len}\widetilde{Z}_0=2$. This provides a $\mathbb{A}^1_s$-family structure on $\widetilde{Z}$.

Exercise 3.29. Show that saturation does not solve the non-flatness issue in Example 3.24.

3.5.1. *Computations for Examples 3.24 and 3.28 and Exercise 3.29.*

```
R=QQ[x,y,s,t];
I=intersect(ideal(x-s,y-s),ideal(x-t,y+t));
J= ideal(x-s,y-s)*ideal(x-t,y+t);
I==J
degree I
degree sub(I,{s=>0,t=>0})
R=QQ[x,y,s];
I=ideal(x,y)*ideal(x-s,y-s);
degree I
degree sub(I,s=>0)
ISat = saturate(I,s)
degree ISat
degree sub(ISat,s=>0)
```

4. Lecture 4

References

1. Marian Aprodu and Jan Nagel, *Koszul cohomology and algebraic geometry*, University Lecture Series, vol. 52, American Mathematical Society, Providence, RI, 2010. MR 2573635
2. Michael Artin, *Lectures on Deformations of Singularities*, Tata Institute of Fundamental Research, Bombay, 1976.
3. Joel Briançon, *Description de Hilb$^n\mathbb{C}\{x,y\}$*, Invent. Math. **41** (1977), 45–89.
4. Dustin A. Cartwright, Daniel Erman, Mauricio Velasco, and Bianca Viray, *Hilbert schemes of 8 points*, Algebra Number Theory **3** (2009), no. 7, 763–795. MR 2579394

5. G. Castelnuovo, *Le trasformazioni generatrici del gruppo cremoniano nel piano*, Atti della R. Acc. delle Scienze di Torino **36** (1901), 861–874.
6. David Eisenbud, *Commutative algebra*, Graduate Texts in Mathematics, vol. 150, Springer-Verlag, New York, 1995, With a view toward algebraic geometry. MR 1322960
7. John Fogarty, *Algebraic families on an algebraic surface*, Amer. J. Math. **90** (1968), 511–521.
8. Michele Graffeo and Giorgio Gubbiotti, *Growth and integrability of some birational maps in dimension three*, Ann. Henri Poincaré **25** (2024), no. 3, 1733–1793. MR 4711236
9. Daniel R. Grayson and Michael E. Stillman, *Macaulay2, a software system for research in algebraic geometry*, Available at http://www2.macaulay2.com.
10. Robin Hartshorne, *Algebraic geometry*, Springer-Verlag, New York-Heidelberg, 1977, Graduate Texts in Mathematics, No. 52.
11. Anthony Iarrobino, *Punctual Hilbert schemes*, Mem. Amer. Math. Soc. **10** (1977), no. 188, viii+112.
12. M. Noether, *Ueber Flächen, welche Schaaren rationaler Curven besitzen*, Math. Ann. **3** (1870), 161–226.

(G. Alecci) Politecnico di Torino, Corso Duca degli Abruzzi 24, 10129, Torino (Italy)
Email address: gessica.alecci@polito.it

(M. Graffeo) SISSA, Via Bonomea 265, 34136, Trieste (Italy)
Email address: mgraffeo@sissa.it